
1 Introduction

We implement an AI v.s. AI Gomoku game, also called Five in a Row. Here it’s played on
a board with size 10*10. Black and white players alternate turns to place a stone of their
color on an empty intersection.
We use minimax search algorithm to depth of three employing alpha-beta cut-off strategy
to address the two players playing against each other.

2 Implementation

There are three files: Board.hs, AI.hs and Main.hs.

2.1 Board.hs

Define a 10*10 Board and each player’s move can denote by a Point. Each point has its
color(Black or White) and its corresponding position (Int, Int).
Key methods are addPoint and checkWin. After each move, we check four directions
whether there is already five same color points in a row. There is no need to check all
the boards, just four lines those includes the latest point.

2.2 AI.hs

Key method here is where to put the current point to realize a game of competition. We
use Minimax algorithm to alternate between the two AI players, the player desires to pick
the move with the maximum score. In turn, the scores for each of available moves are
determined by the opposing player which of its available moves has the minimum score.
Scores are calculated as: score 100000 when 5 in a row; 10000 with 4 in a row; 1000 with
3 in a row and 100 with 2 in a row. Build a tree of depth 3 to compare all possible next
three moves and pick the most favorable one for current move.
Then we improve Minimax by alpha-beta cut-off. Each node has a boundary [alpha,
beta]. alpha means lower boundary and beta means the upper boundary. Each time
when beta ≤ alpha, we no longer search more sub-trees, This is a process of pruning.
We refer to other’s codes when we implement the minmaxAlpha and minmaxBeta methods
but we modified it to fit our data structures.
We used parallel strategies in two places. First, we run the minmax algorithm on each
child board of the current board in the board tree in parallel and choose the one with
highest score as the next move. Second, to rate each board, we use parallel to get the
score of current board in each possible directions. In both cases, we use parMap rdeepseq

as our parallel strategy.
Some steps of alternatively running two AIs on a 10*10 board is shown in the table below.
O and X denotes the two users and denote a vacant place in board.

2.3 Main.hs

Method gameLoop turns on the game by repeatedly calling method moveAI until method
checkWin no longer returns Empty, i.e. one player wins.

3 Performance

We compare performance on one core and that on four cores. We find that running time
on one core is 1.44s and 0.94 on four cores.

Figure 1: Running on one core

Figure 2: Running on four cores

4 Codes

4.1 Main.hs

1 module Main where

2

3 import AI

4 import Board

5 import Data.Char

6

7 gameLoop :: Board -> Color -> [Board] -> IO ()

8 gameLoop board color list

9 | null curPoint = putStrLn "Tie"

10 | checkWin (head curPoint) curBoard == Empty = do

11 putStrLn (show curBoard ++ "\n")

12 gameLoop curBoard (oppositeColor color) (list ++ [curBoard ])

13 | otherwise = do

14 putStrLn (show curBoard ++ "\n")

15 putStrLn (show color ++ "wins")

16 where

17 curPoint = getCurPoint board curBoard

18 curBoard = moveAI board color

19

20 main :: IO ()

21 main = gameLoop initBoard Black []

Listing 1: Main.hs

4.2 Board.hs

1 module Board

2 ( Color (..)

3 , Point (..)

4 , Board (..)

5 , initBoard

6 , oppositeColor

7 , filterBoard

8 , isEmptyBoard

9 , addPoint

10 , isValidPoint

11 , isVacant

12 , checkWin

13 , getCurPoint

14 ) where

15

16 import Data.List

17

18 data Color

19 = Black

20 | White

21 | Empty

22 deriving (Eq)

23

24 instance Show Color where

25 show Black = "X"

26 show White = "O"

27 show Empty = "_"

28

29 data Point =

30 Point

31 { color :: Color

32 , position :: (Int , Int)

33 }

34

35 instance Show Point where

36 show (Point color _) = show color

37

38 instance Eq Point where

39 (Point color1 (x1 , y1)) == (Point color2 (x2 , y2)) = x1 == x2 && y1 ==

y2 && color1 == color2

40

41 instance Ord Point where

42 compare (Point _ (x1 ,y1)) (Point _ (x2,y2)) = compare (x1*10+y1) (x2

*10+y2)

43

44 newtype Board = Board [[ Point]]

45

46 instance Show Board where

47 show (Board points) = intercalate "\n" $ map show points

48

49 instance Eq Board where

50 (Board points1) == (Board points2) = points1 == points2

51

52 initBoard :: Board

53 initBoard = Board points

54 where

55 points = [initRow x 10 | x <- [1 .. 10]]

56 initRow _ 0 = []

57 initRow row col = initRow row (col - 1) ++ [Point Empty (row , col)]

58

59 getPoint :: Board -> (Int , Int) -> Point

60 getPoint (Board points) (x, y) = (points !! (x - 1)) !! (y - 1)

61

62 isValidPoint :: Point -> Bool

63 isValidPoint (Point _ (x, y))

64 | x > 0 && x <= 10 && y > 0 && y <= 10 = True

65 | otherwise = False

66

67 isVacant :: Point -> Board -> Bool

68 isVacant (Point color (x, y)) (Board points) = curColor == Empty

69 where

70 (Point curColor (_, _)) = getPoint (Board points) (x, y)

71

72 addPoint :: Board -> Color -> Int -> Int -> Board

73 addPoint (Board points) color x y

74 | isValidPoint (Point color (x, y)) && isVacant (Point color (x, y)) (

Board points) =

75 add (Point color (x, y)) (Board points)

76 | otherwise = Board points

77

78 add :: Point -> Board -> Board

79 add (Point color (newx , newy)) (Board points) = Board newPoints

80 where

81 newPoints = upperRows ++ (leftCells ++ (Point color (newx , newy) :

rightCells)) : lowerRows

82 (upperRows , thisRow:lowerRows) = splitAt (newx - 1) points

83 (leftCells , _:rightCells) = splitAt (newy - 1) thisRow

84

85 checkWin :: Point -> Board -> Color

86 checkWin (Point color (x, y)) (Board points)

87 | winRow (Point color (x, y)) (Board points) /= 0 ||

88 (winCol (Point color (x, y)) (Board points) /= 0) ||

89 (winDiag (Point color (x, y)) (Board points) /= 0) || (winAntiDiag

(Point color (x, y)) (Board points) /= 0) =

90 color

91 | otherwise = Empty

92

93 checkRow :: [Point] -> Color -> Int -> Int

94 checkRow [] preColor cnt =

95 if cnt == 5

96 then if preColor == Black

97 then 1

98 else 2

99 else 0

100 checkRow (head:xs) preColor cnt

101 | preColor == Empty = checkRow xs color 1

102 | preColor == color && cnt < 4 = checkRow xs color (cnt + 1)

103 | preColor == color && cnt == 4 =

104 if color == Black

105 then 1

106 else 2

107 | otherwise = 0

108 where

109 (Point color _) = head

110

111 getDiag :: Board -> Board

112 getDiag (Board points) = Board $ diagonals points

113

114 getAntiDiag :: Board -> Board

115 getAntiDiag (Board points) = Board $ diagonals (( transpose . reverse)

points)

116

117 diagonals :: [[a]] -> [[a]]

118 diagonals = tail . go []

119 where

120 go b es_ =

121 [h | h:_ <- b] :

122 case es_ of

123 [] -> transpose ts

124 e:es -> go (e : ts) es

125 where

126 ts = [t | _:t <- b]

127

128 winRow :: Point -> Board -> Int

129 winRow (Point color (x, y)) (Board points) = checkRow (newPoints !! (x -

1)) Empty 1

130 where

131 Board newPoints = addPoint (Board points) color x y

132

133 winCol :: Point -> Board -> Int

134 winCol (Point color (x, y)) (Board points) = checkRow (( transpose .

reverse) newPoints !! (y - 1)) Empty 1

135 where

136 Board newPoints = addPoint (Board points) color x y

137

138 winDiag :: Point -> Board -> Int

139 winDiag (Point color (x, y)) (Board points) = checkRow (diagonals

newPoints !! (x + y - 2)) Empty 1

140 where

141 Board newPoints = addPoint (Board points) color x y

142

143 winAntiDiag :: Point -> Board -> Int

144 winAntiDiag (Point color (x, y)) (Board points) =

145 checkRow (diagonals (( transpose . reverse) newPoints) !! (9 - x + y))

Empty 1

146 where

147 Board newPoints = addPoint (Board points) color x y

148

149 isEmptyBoard :: Board -> Bool

150 isEmptyBoard (Board points) = Board points == initBoard

151

152 oppositeColor :: Color -> Color

153 oppositeColor color

154 | color == White = Black

155 | color == Black = White

156 | otherwise = error "Invalid opposite color"

157

158 filterBoard :: Board -> Color -> [Point]

159 filterBoard (Board points) color =

160 [p | rows <- points , p <- rows , isSameColor p]

161 where

162 isSameColor (Point c (_,_)) = c == color

163

164 flatten :: [[a]] -> [a]

165 flatten xs = (\z n -> foldr (flip (foldr z)) n xs) (:) []

166

167 getCurPoint :: Board -> Board -> [Point]

168 getCurPoint (Board points1) (Board points2) = flatten points2 \\ flatten

points1

Listing 2: Board.hs

4.3 AI.hs

1 module AI

2 ( moveAI

3 ) where

4

5 import Board

6 import Control.Parallel.Strategies

7 import Data.List

8 import Data.Maybe

9 import qualified Data.Set as Set

10 import Data.Tree

11

12 minInt :: Int

13 minInt = -(2 ^ 29)

14

15 maxInt :: Int

16 maxInt = 2 ^ 29 - 1

17

18 moveAI :: Board -> Color -> Board

19 moveAI board color

20 | isEmptyBoard board = addPoint board color 1 1

21 | otherwise = bestMove

22 where

23 neighbors = possibleMoves board

24 (Node node children) = buildTree color board neighbors

25 minmax = parMap rdeepseq (minmaxBeta color 3 minInt maxInt) children

26 index = fromJust $ elemIndex (maximum minmax) minmax

27 (Node bestMove _) = children !! index

28

29 buildTree :: Color -> Board -> [Point] -> Tree Board

30 buildTree color board neighbors = Node board $ children neighbors

31 where

32 newNeighbors point =

33 Set.toList $

34 Set.union (Set.fromList (Data.List.delete point neighbors)) (Set.

fromList (stepFromPoint board point))

35 oppoColor = oppositeColor color

36 children [] = []

37 children (Point c (x, y):ns) =

38 buildTree oppoColor (addPoint board color x y) (newNeighbors (Point

c (x, y))) : children ns

39

40 minmaxAlpha :: Color -> Int -> Int -> Int -> Tree Board -> Int

41 minmaxAlpha _ _ alpha _ (Node _ []) = alpha

42 minmaxAlpha color level alpha beta (Node b (x:xs))

43 | level == 0 = curScore

44 | canFinish curScore = curScore

45 | newAlpha >= beta = beta

46 | otherwise = minmaxAlpha color level newAlpha beta (Node b xs)

47 where

48 curScore = scoreBoard b color

49 canFinish score = score > 100000 || score < ( -100000)

50 newAlpha = maximum [alpha , minmaxBeta color (level - 1) alpha beta x]

51

52 minmaxBeta :: Color -> Int -> Int -> Int -> Tree Board -> Int

53 minmaxBeta _ _ _ beta (Node _ []) = beta

54 minmaxBeta color level alpha beta (Node b (x:xs))

55 | level == 0 = curScore

56 | canFinish curScore = curScore

57 | alpha >= newBeta = alpha

58 | otherwise = minmaxBeta color level alpha newBeta (Node b xs)

59 where

60 curScore = scoreBoard b color

61 canFinish score = score > 100000 || score < ( -100000)

62 newBeta = minimum [beta , minmaxAlpha color (level - 1) alpha beta x]

63

64 scoreBoard :: Board -> Color -> Int

65 scoreBoard board color = score (pointsOfColor color) - score (

pointsOfColor $ oppositeColor color)

66 where

67 score points = sum $ map sumScores $ scoreDirections points

68 pointsOfColor = filterBoard board

69

70 sumScores :: [Int] -> Int

71 sumScores [] = 0

72 sumScores (x:xs)

73 | x == 5 = 100000 + sumScores xs

74 | x == 4 = 10000 + sumScores xs

75 | x == 3 = 1000 + sumScores xs

76 | x == 2 = 100 + sumScores xs

77 | otherwise = sumScores xs

78

79 scoreDirections :: [Point] -> [[Int]]

80 scoreDirections [] = [[0]]

81 scoreDirections ps@(point:rest) =

82 parMap

83 rdeepseq

84 (scoreDirection point ps 0)

85 [(xDir , yDir) | xDir <- [0 .. 1], yDir <- [-1 .. 1], not (xDir == 0

&& yDir == (-1)), not (xDir == 0 && yDir == 0)]

86

87 scoreDirection :: Point -> [Point] -> Int -> (Int , Int) -> [Int]

88 scoreDirection _ [] cont (_, _) = [cont]

89 scoreDirection (Point c (x, y)) ps@(Point c1 (x1, y1):rest) cont (xDir ,

yDir)

90 | Point c (x, y) ‘elem ‘ ps =

91 scoreDirection (Point c (x + xDir , y + yDir)) (Data.List.delete (

Point c (x, y)) ps) (cont + 1) (xDir , yDir)

92 | otherwise = cont : scoreDirection (Point c1 (x1 , y1)) rest 1 (xDir ,

yDir)

93

94 possibleMoves :: Board -> [Point]

95 possibleMoves board = Set.toList $ stepBoard board $ filterBoard board

White ++ filterBoard board Black

96

97 stepBoard :: Board -> [Point] -> Set.Set Point

98 stepBoard _ [] = Set.empty

99 stepBoard board (point:rest) = Set.union (Set.fromList (stepFromPoint

board point)) $ stepBoard board rest

100

101 stepFromPoint :: Board -> Point -> [Point]

102 stepFromPoint board (Point _ (x, y)) =

103 [ Point Empty (x + xDir , y + yDir)

104 | xDir <- [-1 .. 1]

105 , yDir <- [-1 .. 1]

106 , not (xDir == 0 && yDir == 0)

107 , isValidPoint (Point Empty (x + xDir , y + yDir))

108 , isVacant (Point Empty (x + xDir , y + yDir)) board

109 ]

Listing 3: AI.hs

5 Reference

1. https://github.com/sowakarol/gomoku-haskell

2. https://github.com/lihongxun945/myblog/issues/14

